

## FSFI Scoring

### SAS Code

```
data fsfi2;
    set fsfi1;
    array FSFIqs(19) FSFI_q1-FSFI_q19;
    array scoring_FSFIqs(19) scoring_FSFI_q1-scoring_FSFI_q19;
```

/\* Questions 3 through 14, plus 17 through 19 have options labeled "No sexual activity" (or similar) and are assigned a score of 0, which has the effect of implying sexual dysfunction for women who are not sexually active. This is a recognized problem with this questionnaire\* which we addressed with changes in our FSFI CRF for MAPP II. Here, we addressed the problem via scoring changes. Specifically, for calculating domain and total scores, responses of 0 were changed to missing

\*For more information, refer to:

Heino F. L. Meyer-Bahlburg & Curtis Dolezal (2007) The Female Sexual Function Index: A Methodological Critique and Suggestions for Improvement, Journal of Sex & Marital Therapy, 33:3, 217-224, DOI: 10.1080/00926230701267852 \*/

```
fsfi_numzeros = 0;
do i = 1 to 19;
    if fsfiqs(i) = 0 then do;
        scoring_fsfiqs(i) = .;
        fsfi_numzeros = fsfi_numzeros + 1;
    end;
    else scoring_fsfiqs(i) = fsfiqs(i);
end;

drop i;
label fsfi_numzeros = "Number of No Sexual Activity Responses (0-15)";
run;
```

/\* Create Domain Scores and Total Score. Note that domain scores are adjusted based on the number of questions they include so that they are all comparable and have equal weight in calculating the total score \*/

```
data fsfi3;
    set fsfi2;
```

/\* Domain 1: Desire - qs 1 and 2 \*/

```
raw_fsfi_desire = scoring_fsfi_q1 + scoring_fsfi_q2;
fsfi_desire = raw_fsfi_desire*0.6;
```

```
/* Domain 2: Arousal - qs 3, 4, 5, 6 */
```

```
non_miss_arousal = n(of scoring_fsfi_q3-scoring_fsfi_q6); /* impute if 3 or more non-missing */  
if non_miss_arousal ge 3  
  then raw_fsfi_arousal = (sum (of scoring_fsfi_q3-scoring_fsfi_q6))*4/non_miss_arousal;  
fsfi_arousal = raw_fsfi_arousal*0.3;
```

```
/* Domain 3: lubrication - qs 7, 8, 9, and 10 */
```

```
non_miss_lub = n(of scoring_fsfi_q7-scoring_fsfi_q10); /* impute if 3 or more non-missing */  
if non_miss_lub ge 3 then raw_fsfi_lub = (sum (of scoring_fsfi_q7-scoring_fsfi_q10))*4/non_miss_lub;  
fsfi_lubrication = raw_fsfi_lub*0.3;
```

```
/* Domain 4: orgasm - qs 11, 12, 13 */
```

```
non_miss_orgasm = n(of scoring_fsfi_q11-scoring_fsfi_q13); /* impute if 2 or more non-missing */  
if non_miss_orgasm ge 2 then  
  raw_fsfi_orgasm = (sum (of scoring_fsfi_q11-scoring_fsfi_q13))*3/non_miss_orgasm;  
fsfi_orgasm = raw_fsfi_orgasm*0.4;
```

```
/* Domain 5: satisfaction - qs 14, 15, 16 */
```

```
non_miss_satisfaction = n(of scoring_fsfi_q14-scoring_fsfi_q16); /* impute if 2 or more non-missing */  
if non_miss_satisfaction ge 2 then  
  raw_fsfi_satisfaction = (sum (of scoring_fsfi_q14-scoring_fsfi_q16))*3/non_miss_satisfaction;  
fsfi_satisfaction = raw_fsfi_satisfaction*0.4;
```

```
/* Domain 6: pain - qs 17, 18, 19 */
```

```
non_miss_pain = n(of scoring_fsfi_q17-scoring_fsfi_q19); /* impute if 2 or more non-missing */  
if non_miss_pain ge 2 then  
  raw_fsfi_pain = (sum (of scoring_fsfi_q17-scoring_fsfi_q19))*3/non_miss_pain;  
fsfi_pain = raw_fsfi_pain*0.4;
```

```
/* Total */
```

```
non_miss_domains = n(fsfi_desire, fsfi_arousal, fsfi_lubrication, fsfi_orgasm,  
  fsfi_satisfaction, fsfi_pain);
```

```
if non_miss_domains ge 4 then fsfi_total = (sum (FSFI_desire, FSFI_arousal, FSFI_lubrication,  
  FSFI_orgasm, FSFI_satisfaction, FSFI_pain))*6/non_miss_domains;
```

```
drop raw: scoring;;  
run;
```